

ME 406

Using Eigenvector Methods with *Mathematica* to Solve Linear Autonomous Systems of First Order Differential Equations

■ 1. Introduction

In this notebook, we use the methods of linear algebra -- specifically eigenvector and eigenvalue analysis -- to solve systems of linear autonomous ordinary differential equations. Although the *Mathematica* routines DSolve and NDSolve could be used to attack these problems directly, we do not use them here. Our purpose is to make clear the underlying linear algebra, and to use *Mathematica* to do all of the calculations. The canonical problem under consideration is the following:

$$\dot{\mathbf{X}} = \mathbf{A}\mathbf{X} + \mathbf{b} , \quad (1)$$

$$\mathbf{X}(0) = \mathbf{X}_0 , \quad (2)$$

where \mathbf{A} is a constant $n \times n$ matrix, \mathbf{b} is a constant $n \times 1$ vector, and \mathbf{X}_0 , the initial vector, is a given $n \times 1$ vector.

The general approach to this problem is the following. We first find a particular solution, which is defined to be any solution of equation (1). We call the particular solution \mathbf{X}_p . The general solution \mathbf{X}_g of the equation is then

$$\mathbf{X}_g = \mathbf{X}_h + \mathbf{X}_p , \quad (3)$$

where \mathbf{X}_h is the most general solution of the homogeneous equation:

$$\dot{\mathbf{X}}_h = \mathbf{A}\mathbf{X}_h . \quad (4)$$

The homogenous equation (4) has n linearly independent solutions. We call them $\mathbf{X}^{(1)}$, $\mathbf{X}^{(2)}$, \dots , $\mathbf{X}^{(n)}$. The most general \mathbf{X}_h is a linear combination of these. To solve the initial-value problem, we form the general solution \mathbf{X}_g in terms of n constants a_1, a_2, \dots, a_n :

$$\mathbf{X}_g = \mathbf{X}_p + a_1 \mathbf{X}^{(1)} + a_2 \mathbf{X}^{(2)} + \dots + a_n \mathbf{X}^{(n)} . \quad (5)$$

We now impose the initial condition (2) on the solution (5). This gives the following set of linear equations to solve for the coefficients a_i :

$$a_1 \mathbf{X}^{(1)} + a_2 \mathbf{X}^{(2)} + \dots + a_n \mathbf{X}^{(n)} = \mathbf{X}_0 - \mathbf{X}_p \text{ at } t = 0. \quad (6)$$

The linear independence of the vectors $\mathbf{X}^{(i)}$ guarantees that the matrix in the above equations is nonsingular and hence the solution for the coefficients a_i is unique.

The rest of this notebook provides the details in carrying this out, and shows how to use *Mathematica* to advantage at each step. We will begin with a brief review of matrix manipulations in *Mathematica*. Then we will consider the problem of finding the particular solution. After that, we will solve the homogeneous equation, in the following sequence of cases of increasing difficulty: distinct real eigenvalues; distinct complex eigenvalues; repeated eigenvalues. Detailed examples will be done at each step.

■ 2. Basic Matrix Manipulations in Mathematica

In *Mathematica*, a matrix is a list of lists. Each component list is a row of the matrix. As an example, we define a matrix named A for *Mathematica*, and then use `MatrixForm` to print it out in traditional form.

```
A = {{1, -2, 3}, {-2, -1, 4}, {3, 4, 5}};
```

```
MatrixForm[A]
```

$$\begin{pmatrix} 1 & -2 & 3 \\ -2 & -1 & 4 \\ 3 & 4 & 5 \end{pmatrix}$$

One way to find out whether A is singular is to compute the determinant.

```
Det[A]
```

```
-80
```

Because the determinant is not zero, A is not singular, and thus has an inverse.

```
B = Inverse[A]
```

$$\left\{ \left\{ \frac{21}{80}, -\frac{11}{40}, \frac{1}{16} \right\}, \left\{ -\frac{11}{40}, \frac{1}{20}, \frac{1}{8} \right\}, \left\{ \frac{1}{16}, \frac{1}{8}, \frac{1}{16} \right\} \right\}$$

We check the inverse by forming the product of A and the inverse of A. The product should be the identity matrix. Matrix products are indicated by a period.

```
A.B
```

```
{{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}
```

```
B.A
```

```
{{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}
```

We can use this inverse to solve a set of linear algebraic equations in which A is the coefficient matrix. Consider the equations

$$x - 2y + 3z = 1,$$

$$-2x - y + 4z = 0,$$

(7)

$$3x + 4y + 5z = -1.$$

The coefficient matrix is the matrix A , and the right-hand side is the vector

$$\mathbf{b} = \{1, 0, -1\};$$

We can use the inverse we have already calculated to solve these equations. We call the solution sol .

$$\text{sol} = \mathbf{B}.\mathbf{b}$$

$$\left\{ \frac{1}{5}, -\frac{2}{5}, 0 \right\}$$

We check this:

$$\mathbf{A}.\text{sol} - \mathbf{b}$$

$$\{0, 0, 0\}$$

An alternative method of solution is to use *Mathematica's* `LinearSolve`.

$$\text{sol2} = \text{LinearSolve}[\mathbf{A}, \mathbf{b}]$$

$$\left\{ \frac{1}{5}, -\frac{2}{5}, 0 \right\}$$

We get the same result.

In the above calculations, all the elements of A were integers, so *Mathematica* did an exact calculation. If one or more elements of A are written as real numbers, *Mathematica* will do a numerical rather than exact calculation.

$$\mathbf{Amod} = \{\{1., -2, 3\}, \{-2, -1, 4\}, \{3, 4, 5\}\};$$

$$\text{Inverse}[\mathbf{Amod}]$$

$$\{\{0.2625, -0.275, 0.0625\}, \\ \{-0.275, 0.05, 0.125\}, \{0.0625, 0.125, 0.0625\}\}$$

For the present calculations, it doesn't matter much which way we do them. For larger matrices and more difficult calculations, such as eigenanalysis, it is crucial to do the calculations numerically. We shall see an example of this shortly.

The determination of eigenvalues and eigenvectors is the central linear algebra calculation for solving systems of first-order linear autonomous differential equations. Given a square matrix \mathbf{A} , we say that a non-zero vector \mathbf{c} is an eigenvector of \mathbf{A} with eigenvalue λ if $\mathbf{A}\mathbf{c} = \lambda\mathbf{c}$. *Mathematica* has a lot of built-in power to find eigenvectors and eigenvalues. We go back to our matrix \mathbf{A} and use *Mathematica* to find its eigenvalue.

eigvals = Eigenvalues [A]

$$\left\{ \frac{5}{3} + \frac{23 \cdot 5^{2/3}}{3 \left(-56 + 3 i \sqrt{6411}\right)^{1/3}} + \frac{1}{3} \left(5 \left(-56 + 3 i \sqrt{6411}\right)\right)^{1/3}, \right.$$

$$\frac{5}{3} - \frac{23 \cdot 5^{2/3} (1 + i \sqrt{3})}{6 \left(-56 + 3 i \sqrt{6411}\right)^{1/3}} - \frac{1}{6} (1 - i \sqrt{3}) \left(5 \left(-56 + 3 i \sqrt{6411}\right)\right)^{1/3},$$

$$\left. \frac{5}{3} - \frac{23 \cdot 5^{2/3} (1 - i \sqrt{3})}{6 \left(-56 + 3 i \sqrt{6411}\right)^{1/3}} - \frac{1}{6} (1 + i \sqrt{3}) \left(5 \left(-56 + 3 i \sqrt{6411}\right)\right)^{1/3} \right\}$$

We are getting a graphic reminder about the difference between numerical and exact. Would you guess from the above expression that the eigenvalues of **A** are real? Because **A** is symmetric it must have real eigenvalues. Let's evaluate the above numerically.

N[eigvals]

$$\{7.56738 - 1.33227 \times 10^{-15} i,$$

$$-4.77934 - 1.11022 \times 10^{-15} i, 2.21196 + 2.66454 \times 10^{-15} i\}$$

We see that they are real, plus a tiny imaginary part due to numerical error. It is much easier just to get the eigenvalues numerically in the first place by using the modified form **Amod**, which contains a real number among its entries.

eigvals = Eigenvalues [Amod]

$$\{7.56738, -4.77934, 2.21196\}$$

The eigenvectors are also easy to get. The *Mathematica* command is **Eigenvectors**.

Eigenvectors [Amod]

$$\{\{0.301125, 0.344798, 0.889066\},$$

$$\{-0.492759, -0.74195, 0.45464\}, \{0.816402, -0.574999, -0.0535175\}\}$$

To get all of this information at once, we use **Eigensystem**.

Eigensystem [Amod]

$$\{\{7.56738, -4.77934, 2.21196\},$$

$$\{\{0.301125, 0.344798, 0.889066\}, \{-0.492759, -0.74195, 0.45464\},$$

$$\{0.816402, -0.574999, -0.0535175\}\}\}$$

Let's repeat this command and assign the eigenvalues to the symbols λ_1 , λ_2 , λ_3 , and the eigenvectors to the symbols v_1 , v_2 , v_3 .

```
{ $\{\lambda_1, \lambda_2, \lambda_3\}, \{v_1, v_2, v_3\}\} = \text{Eigensystem}[A_{\text{mod}}]$ 
{
  {7.56738, -4.77934, 2.21196},
  {{0.301125, 0.344798, 0.889066}, {-0.492759, -0.74195, 0.45464},
  {0.816402, -0.574999, -0.0535175}}
```

We sample these assignments:

```
 $\lambda_1$ 
7.56738
```

```
 $v_1$ 
{0.301125, 0.344798, 0.889066}
```

Let's check that these eigenvectors and eigenvalues work.

```
 $A_{\text{mod}} \cdot v_1 - \lambda_1 * v_1$ 
{-4.44089  $\times 10^{-16}$ , 8.88178  $\times 10^{-16}$ , -8.88178  $\times 10^{-16}$ }
```

```
 $A_{\text{mod}} \cdot v_2 - \lambda_2 * v_2$ 
{1.33227  $\times 10^{-15}$ , 4.88498  $\times 10^{-15}$ , -8.88178  $\times 10^{-16}$ }
```

```
 $A_{\text{mod}} \cdot v_3 - \lambda_3 * v_3$ 
{6.66134  $\times 10^{-16}$ , 2.22045  $\times 10^{-15}$ , -2.67841  $\times 10^{-15}$ }
```

Close enough!

There are three other *Mathematica* matrix commands which are useful in the present context. The first is `MatrixPower`, which raises a matrix to an integer power. For example, the command below raises the matrix `Amod` to the third power.

```
 $A_{\text{mod}3} = \text{MatrixPower}[A_{\text{mod}}, 3]$ 
{{20., 0., 140.}, {0., -5., 170.}, {140., 170., 320.}}
```

As a check, we can verify that `Amod3` has the same eigenvectors as `Amod`, but with the eigenvalues cubed. We do this for `v1` and `λ_1` .

```
 $A_{\text{mod}3} \cdot v_1 - (\lambda_1)^3 \cdot v_1$ 
{-5.68434  $\times 10^{-14}$ , 0., -1.7053  $\times 10^{-13}$ }
```

The second command is `NullSpace`. If a matrix \mathbf{M} is singular, then there are some vectors \mathbf{v} for which $\mathbf{M} \cdot \mathbf{v} = 0$. The set of all such vectors is a vector space, and it is called the nullspace of \mathbf{M} . An alternative way of

expressing this: the nullspace of \mathbf{M} is the set of all solutions of the homogeneous equations $\mathbf{M}\cdot\mathbf{v} = 0$. The command `NullSpace[M]` returns a basis for the nullspace. For example,

```
NullSpace[Amod]
```

```
{}
```

Because \mathbf{Amod} is not singular, there are no vectors in the nullspace other than the zero vector. A more interesting example follows. In solving the eigenvalue problem $\mathbf{A}\mathbf{c} = \lambda\mathbf{c}$, one can rewrite it as $(\mathbf{A} - \lambda\mathbf{I})\mathbf{c} = 0$, where \mathbf{I} is the identity matrix. Thus the matrix $\mathbf{A} - \lambda\mathbf{I}$ is singular and has a non-trivial nullspace. In fact the nullspace is the eigenvector associated with λ . Let's try that. We define the identity matrix first, using the *Mathematica* function `IdentityMatrix[n]`, which gives the $n \times n$ identity matrix.

```
Id = IdentityMatrix[3]
```

```
{{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}
```

```
NullSpace[Amod - λ1 * Id]
```

```
{{-0.301125, -0.344798, -0.889066}}
```

```
v1
```

```
{0.301125, 0.344798, 0.889066}
```

Apart from a non-essential minus sign, we get the same thing, as we must.

The third command is `CharacteristicPolynomial`, which returns the polynomial $\text{Det}[\mathbf{A} - \lambda\mathbf{I}]$. The roots of this polynomial are the eigenvalues of \mathbf{A} . Let's try it.

```
poly = CharacteristicPolynomial[Amod, λ]
```

```
(-4.77934 - λ) (2.21196 - λ) (7.56738 - λ)
```

It is convenient that *Mathematica* has returned the polynomial in factored form, from which we see that the roots are indeed the eigenvalues calculated earlier. We also could accomplish all of this "by hand" in the following way.

```
poly2 = Det[Amod - λ * Id]
```

```
-80. + 30. λ + 5. λ2 - λ3
```

```
Solve[poly2 == 0, λ]
```

```
{{λ → -4.77934}, {λ → 2.21196}, {λ → 7.56738}}
```

■ 3. Finding the Particular Solution

We seek any solution of equation (1) -- the simpler the better. Nothing could be simpler than an equilibrium solution (i.e, $\mathbf{X} = \text{constant}$), so we look for one, and we find that if the matrix \mathbf{A} is nonsingular, then

$$\mathbf{X}_p = -\mathbf{A}^{-1} \mathbf{b} . \quad (8)$$

which is an equilibrium solution. If \mathbf{A} is singular, equation (8) is not a valid solution, and the situation is more complicated. It turns out that there may or may not be an equilibrium solution, depending on both \mathbf{A} and \mathbf{b} . Of course there is always a particular solution, whether or not there is an equilibrium solution. It is an interesting excursion into linear algebra to sort all of this out, but it would be a distraction for us, because our main purpose here is to develop the tools for stability analysis, and it is the homogeneous equation that provides what we need. We will not consider any further the case of singular \mathbf{A} here. We just note that in all cases, one can always find a particular solution in the form of a vector with components which are polynomials in t .

■ 4. Solving the Homogeneous Equation

■ 4.1 Basic Concepts

We attack the homogeneous equation (4), using the basic theory of eigenvalues and eigenvectors. Because the equation is homogeneous and constant coefficient, we would guess from our experience in elementary differential equations that there will be solutions of the form

$$\mathbf{X} = e^{\lambda t} \mathbf{c} , \quad (9)$$

where the constant λ and the constant vector \mathbf{c} are to be determined. We substitute this form into equation (4) to get

$$\mathbf{A}\mathbf{c} = \lambda\mathbf{c} , \quad (10)$$

which is the matrix eigenvalue problem for \mathbf{A} . As we noted earlier, the condition for (11) to have a nontrivial solution is that

$$p(\lambda) = 0 , \text{ where } p(\lambda) = \det(\mathbf{A} - \lambda\mathbf{I}) . \quad (11)$$

Here $p(\lambda)$ is the characteristic polynomial and is of degree n , where n is the order of the system (i.e., \mathbf{A} is $n \times n$, \mathbf{X} is $n \times 1$). Thus p will have n roots. For each root λ , we can generate a solution of the form (9). We need n such linearly independent solutions for a solution basis of (4), and this is consistent with the fact that p has n roots. Although this all does work, there are complications. In the simplest case, p has n distinct real roots, and thus n independent solutions of the form (10). This case is dealt with in section 4.2 below. Complex roots present a minor difficulty, mainly because in general we seek real-valued solutions of these real equations. It is not hard to convert the complex roots into real solutions, and this is done in section 4.3 below. The most troublesome case is that of repeated roots. This requires the introduction of the concept of generalized eigenvectors. This is done in section 4.4, and some practical computational tools are given there also.

■ 4.2 Real Distinct Eigenvalues

This case is straightforward. By assumption, the eigenvalues $\lambda_1, \lambda_2, \dots, \lambda_n$ are real and distinct, and the associated eigenvectors are also real. Each pair (λ, \mathbf{c}) gives rise to a real solution of the form (10). The general solution of the the homogeneous equation (4) is then

$$\mathbf{X}_h = a_1 \mathbf{c}^{(1)} e^{\lambda_1 t} + a_2 \mathbf{c}^{(2)} e^{\lambda_2 t} + \dots + a_n \mathbf{c}^{(n)} e^{\lambda_n t} . \quad (12)$$

The only possible remaining question is whether the solutions $\mathbf{c}^{(i)} e^{\lambda_i t}$ are linearly independent. For simplicity in writing, let $b_i = a_i e^{\lambda_i t}$. Then if the solutions are linearly dependent, we must have

$$b_1 \mathbf{c}^{(1)} + b_2 \mathbf{c}^{(2)} + \dots + b_n \mathbf{c}^{(n)} = 0 \quad (13)$$

for not all $b_i = 0$. Operate on the above equation with the matrix

$$(\mathbf{A} - \lambda_2 \mathbf{I})(\mathbf{A} - \lambda_3 \mathbf{I}) \dots (\mathbf{A} - \lambda_n \mathbf{I}) . \quad (14)$$

The result is

$$a_1 (\lambda_1 - \lambda_2)(\lambda_1 - \lambda_3) \dots (\lambda_1 - \lambda_n) \mathbf{c}^{(1)} = 0 .$$

Because the eigenvalues are distinct and because $\mathbf{c}^{(1)} \neq 0$, it follows that $a_1 = 0$. In the same way, one shows that all $a_i = 0$. Hence only the trivial combination vanishes, and the solutions $\mathbf{c}^{(i)} e^{\lambda_i t}$ are linearly independent.

Consider an example. We use our earlier matrix **Amod**.

MatrixForm[Amod]

$$\begin{pmatrix} 1 & -2 & 3 \\ -2 & -1 & 4 \\ 3 & 4 & 5 \end{pmatrix}$$

We solve the system (1) with $\mathbf{A} = \mathbf{Amod}$, and

$$\mathbf{b} = \{-3, 4, -5\};$$

We will impose the initial condition

$$\mathbf{x}_0 = \{1, 1, 1\};$$

We start by finding the particular solution.

$$\mathbf{x}_p = -\text{Inverse}[\mathbf{Amod}].\mathbf{b}$$

$$\{2.2, -0.4, 0.\}$$

We check the eigenvalues of **Amod** to see if they are distinct.

Eigenvalues[Amod]

$$\{7.56738, -4.77934, 2.21196\}$$

They are distinct. Now we find the eigenvectors, and name the eigenvalues and eigenvectors:

```

{{λ1, λ2, λ3}, {v1, v2, v3}} = Eigensystem[Amod]

{{7.56738, -4.77934, 2.21196},
 {{0.301125, 0.344798, 0.889066}, {-0.492759, -0.74195, 0.45464},
 {0.816402, -0.574999, -0.0535175}}}}

```

We form the homogeneous solution.

```

Xh = a1 * v1 * Exp[λ1 * t] + a2 * v2 * Exp[λ2 * t] + a3 * v3 * Exp[λ3 * t]

{-0.492759 a2 e-4.77934 t + 0.816402 a3 e2.21196 t + 0.301125 a1 e7.56738 t,
 -0.74195 a2 e-4.77934 t - 0.574999 a3 e2.21196 t + 0.344798 a1 e7.56738 t,
 0.45464 a2 e-4.77934 t - 0.0535175 a3 e2.21196 t + 0.889066 a1 e7.56738 t}

```

The general solution is the sum of \mathbf{Xh} and \mathbf{Xp} . We impose the initial conditions on the general solution to determine the values of the constants a_i .

```

coeff = Solve[(Xh + Xp) /. t -> 0 == X0, {a1, a2, a3}]

{{a1 -> 1.01043, a2 -> 0.00722087, a3 -> -1.8382}}

```

Now we use a replacement operation to put these values into the formula and thereby generate the solution, which we call \mathbf{Xsole} .

```

Xsole = (Xh + Xp) /. Thread[Flatten[coeff]]

{2.2 - 0.00355815 e-4.77934 t - 1.50071 e2.21196 t + 0.304267 e7.56738 t,
 -0.4 - 0.00535753 e-4.77934 t + 1.05696 e2.21196 t + 0.348395 e7.56738 t,
 0. + 0.0032829 e-4.77934 t + 0.0983758 e2.21196 t + 0.898341 e7.56738 t}

```

Let's check it.

```

Simplify[D[Xsole, t] - Amod.Xsole - b]

{-4.44089 × 10-16 - 6.93889 × 10-18 e-4.77934 t + 1.33227 × 10-15 e2.21196 t +
 0. e7.56738 t, 0. - 3.46945 × 10-17 e-4.77934 t + 4.44089 × 10-15 e2.21196 t -
 8.88178 × 10-16 e7.56738 t, 0. + 3.46945 × 10-18 e-4.77934 t -
 5.32907 × 10-15 e2.21196 t + 8.88178 × 10-16 e7.56738 t}

```

We see very small residuals associated with numerical error, but they are essentially zero. Now the initial condition:

```

(Xsole /. t -> 0) - X0

{-3.33067 × 10-16, 2.22045 × 10-16, -1.11022 × 10-16}

```

This checks.

■ 4.3 Distinct Complex Eigenvalues

If λ is a complex eigenvalue, then the eigenvector \mathbf{c} will be complex, and the solution $\mathbf{X} = \mathbf{c}e^{\lambda t}$ is complex-valued. If we don't mind having complex valued solutions in our solutions basis, then the solution obtained in the above section works just as well for distinct complex eigenvalues. However, we generally prefer a real-valued solution basis for our real equations. In this section, we see how to obtain such a basis.

Because the coefficient matrix \mathbf{A} is real, it follows that if \mathbf{X} is a complex-valued solution of equation (4), then both the real and imaginary parts of \mathbf{X} are also solutions. In this way, we can generate two real-valued solutions from a pair of complex conjugate eigenvalues. The separation of a complex valued solution into real and imaginary parts requires a few special *Mathematica* commands and options. The best way to see how all of this is done is to work through an example in detail. We consider equation (4) with

```
A = {{8, 3, 15}, {0, -1, 0}, {-3, -3, -4}};
```

We check the eigenvalues of \mathbf{A} .

```
Eigenvalues[A]
```

```
{-1, 2 - 3 i, 2 + 3 i}
```

We have three distinct eigenvalues, with two of them being complex. We construct the eigenvectors, and name the eigenvalues and eigenvectors.

```
{{λ1, λ2, λ3}, {v1, v2, v3}} = Eigensystem[A]
```

```
{{-1, 2 - 3 i, 2 + 3 i}, {{-2, 1, 1}, {-2 + i, 0, 1}, {-2 - i, 0, 1}}}
```

The real-valued solution is

```
X1 = v1 * Exp[λ1 * t]
```

```
{-2 e-t, e-t, e-t}
```

The first complex-valued solution is

```
C2 = v2 * Exp[λ2 * t]
```

```
{(-2 + i) e(2-3 i) t, 0, e(2-3 i) t}
```

Now we will parlay this into two real-valued solutions, by taking the real and imaginary parts. To do this, you must know that the command `ComplexExpand` is necessary to force *Mathematica* to split things into real and imaginary parts. The command for the real part is `Re`, and the command for the imaginary part is `Im`. There is one more difficulty -- a logical one that we can't blame on *Mathematica*. *Mathematica* has no way of knowing whether our independent variable t is real or complex. We know that it is real, but we must tell *Mathematica* that. The standard way of passing such information to *Mathematica* in *Mathematica* 4 is as an assumption in a `Simplify` command. In this case the relevant assumption is that t is real, which in *Mathematica* language is $t \in \text{Reals}$. Note that the symbol \in is NOT epsilon (ϵ or ε), but is the set inclusion symbol on the basic typesetting palette (in the block with the integral signs on the first row). Now we are finally ready to generate our two additional solutions.

```
X2 = Simplify[Re[ComplexExpand[C2]], t ∈ Reals]
```

```
{e2t (-2 Cos[3 t] + Sin[3 t]), 0, e2t Cos[3 t]}
```

```
X3 = Simplify[Im[ComplexExpand[C2]], t ∈ Reals]
```

```
{e2t (Cos[3 t] + 2 Sin[3 t]), 0, -e2t Sin[3 t]}
```

The general solution for this homogeneous problem is then

```
Xgen = a1 * X1 + a2 * X2 + a3 * X3
```

```
{-2 a1 e-t + a2 e2t (-2 Cos[3 t] + Sin[3 t]) + a3 e2t (Cos[3 t] + 2 Sin[3 t]),  
a1 e-t, a1 e-t + a2 e2t Cos[3 t] - a3 e2t Sin[3 t]}
```

We can use this to solve any particular initial value problem. You might want to think about how we were able to generate a complete solution basis without ever using the third eigenvalue or eigenvector.

■ 4.4 Repeated Eigenvalues

In this last section, we tackle the more intricate case of repeated eigenvalues. We will not repeat all of the theory given in class, but all of the computational tools are given here, along with some summary of the basic theory. As we discussed in class, with an eigenvalue λ of \mathbf{A} of multiplicity m , our computational effort centers on finding the generalized eigenvectors -- namely the solutions \mathbf{v} of

$$(\mathbf{A} - \lambda \mathbf{I})^m \mathbf{v} = 0 . \quad (15)$$

Linear algebra guarantees us m linearly independent solutions. Each such solution \mathbf{v} may be transformed into a solution \mathbf{X} of the differential equation (4), by using the formula

$$\mathbf{X} = \left\{ \mathbf{I} + t(\mathbf{A} - \lambda \mathbf{I}) + \frac{t^2}{2!} (\mathbf{A} - \lambda \mathbf{I})^2 + \frac{t^3}{3!} (\mathbf{A} - \lambda \mathbf{I})^3 + \dots + \frac{t^{m-1}}{(m-1)!} (\mathbf{A} - \lambda \mathbf{I})^{m-1} \right\} \mathbf{v} e^{\lambda t} . \quad (16)$$

A *Mathematica* function which carries out this transformation from \mathbf{v} to \mathbf{X} is given below, following a definition of the identity matrix of dimension `ndim`.

```
Id := IdentityMatrix[ndim]
```

```
solmake[A_, λ_, m_, v_, t_] := Module[{ans, i, term}, ans = Id;  
Sum[term = MatrixPower[(A - λ * Id), i] * ((t ^ i) / (i!));  
ans = ans + term, {i, 1, m - 1}]; ans = Exp[λ * t] * (ans.v); ans]
```

The principle computational problem is the solution of (15). After the matrix $(\mathbf{A} - \lambda \mathbf{I})^m$ is formed by using `MatrixPower`, we ask for its `NullSpace`, and that will return a set of m linearly independent solutions of (16). We illustrate all of this by a number of examples.

■ Example 1

We try all of this on the following simple example.

```
A = {{1, 1, 0}, {0, 1, 1}, {0, 0, 1}};
```

```
ndim = 3;
```

The first step is to get the eigenvalues of A.

```
Eigenvalues [A]
```

```
{1, 1, 1}
```

On the off-chance that A might have three linearly independent eigenvectors, we ask for them.

```
Eigenvectors [A]
```

```
{{1, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

No such luck. Only one eigenvector, so we must use our technique of generalized eigenvectors. The multiplicity is 3, so we form $(A - \lambda I)^3$.

```
m = 3;
```

```
λ = 1;
```

```
M = MatrixPower [ (A - λ * Id) , m]
```

```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

The generalized eigenvectors span the nullspace of M. In this case M is the zero matrix, so every vector is in the nullspace of M, and we may choose our three generalized eigenvectors at convenience. We choose them as

```
e1 = {1, 0, 0};
```

```
e2 = {0, 1, 0};
```

```
e3 = {0, 0, 1};
```

Now we construct the solutions associated with each of these.

```
Clear [x1, x2, x3];
```

```
x1[t_] = solmake[A, λ, m, e1, t]
```

```
{et, 0, 0}
```

```
x2[t_] = solmake[A, λ, m, e2, t]
```

```
{et t, et, 0}
```

```
x3[t_] = solmake[A, λ, m, e3, t]
```

```
{ $\frac{e^t t^2}{2}$ , et t, et}
```

These form a solution basis for the homogeneous equation.

■ Example 2

Let's consider an example of a fifth order system.

```
A = {{1, 1, 1, 1, 1}, {0, 1, 1, 1, 1},
      {0, 0, 1, 1, 1}, {0, 0, 0, 1, 1}, {0, 0, 0, 0, 1}};
```

```
Eigenvalues[A]
```

```
{1, 1, 1, 1, 1}
```

```
Eigenvectors[A]
```

```
{{1, 0, 0, 0, 0}, {0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0}, {0, 0, 0, 0, 0}, {0, 0, 0, 0, 0}}
```

We have an eigenvalue of multiplicity 5, with only one linearly independent eigenvector. Thus we must use generalized eigenvectors. We have

```
 $\lambda = 1;$ 
```

```
m = 5;
```

```
ndim = 5;
```

```
M = MatrixPower[A -  $\lambda$  * Id, 5]
```

```
{{0, 0, 0, 0, 0}, {0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0}, {0, 0, 0, 0, 0}, {0, 0, 0, 0, 0}}
```

Again this is the zero matrix, so again we may any choice that is convenient for the five linearly independent generalized eigenvectors.

```
e1 = {1, 0, 0, 0, 0};
```

```
e2 = {0, 1, 0, 0, 0};
```

```
e3 = {0, 0, 1, 0, 0};
```

```
e4 = {0, 0, 0, 1, 0};
```

```
e5 = {0, 0, 0, 0, 1};
```

Now we use solmake to construct our five linearly independent solutions.

```
Clear[x1, x2, x3, x4, x5];
```

```
x1[t_] = solmake[A,  $\lambda$ , m, e1, t]
```

```
{et, 0, 0, 0, 0}
```

```
X2[t_] = solmake[A, λ, m, e2, t]
```

```
{et t, et, 0, 0, 0}
```

```
X3[t_] = solmake[A, λ, m, e3, t]
```

```
{et (t +  $\frac{t^2}{2}$ ), et t, et, 0, 0}
```

```
X4[t_] = solmake[A, λ, m, e4, t]
```

```
{et (t + t2 +  $\frac{t^3}{6}$ ), et (t +  $\frac{t^2}{2}$ ), et t, et, 0}
```

```
X5[t_] = solmake[A, λ, m, e5, t]
```

```
{et (t +  $\frac{3t^2}{2}$  +  $\frac{t^3}{2}$  +  $\frac{t^4}{24}$ ), et (t + t2 +  $\frac{t^3}{6}$ ), et (t +  $\frac{t^2}{2}$ ), et t, et}
```

■ Example 3

We do another example, in which the generalized eigenvectors are more interesting. This example is from **Elementary Differential Equations**, 3rd edition, C.H. Edwards and D.E. Penney, 1994, p. 436.

```
A = {{0, 0, 1, 0}, {0, 0, 0, 1}, {-2, 2, -3, 1}, {2, -2, 1, -3}};
```

```
ndim = 4;
```

```
Eigenvalues[A]
```

```
{-2, -2, -2, 0}
```

We get two distinct eigenvalues, one of multiplicity 3. We look at the eigenvectors:

```
Eigensystem[A]
```

```
{{-2, -2, -2, 0},  
{0, -1, 0, 2}, {-1, 0, 2, 0}, {0, 0, 0, 0}, {1, 1, 0, 0}}
```

We get two linearly independent eigenvectors for $\lambda = -2$, but we still need a third solution for this case, so we use our generalized eigenvectors. First we find the solution associated with the non-degenerate eigenvalue $\lambda = 0$.

```
λ = 0;
```

```
ndim = 4;
```

```
m = 1;
```

```
v = {1, 1, 0, 0};
```

```

Clear[X1, X2, X3, X4, Xgen, Xsole];
X1[t_] = solmake[A, λ, m, v, t]
{1, 1, 0, 0}

```

Now we find the generalized eigenvectors associated with $\lambda = -2$.

```

λ = -2;
m = 3;
M = MatrixPower[(A - λ * Id), 3]
{{4, 4, 2, 2}, {4, 4, 2, 2}, {0, 0, 0, 0}, {0, 0, 0, 0}}

```

By asking for the nullspace of M, we will get a set of three linearly independent generalized eigenvectors.

```

ns = NullSpace[M]
{{-1, 0, 0, 2}, {-1, 0, 2, 0}, {-1, 1, 0, 0}}

```

Our three generalized eigenvectors are

```

e2 = ns[[1]]
{-1, 0, 0, 2}

e3 = ns[[2]]
{-1, 0, 2, 0}

e4 = ns[[3]]
{-1, 1, 0, 0}

```

Now we construct a solution from each of these.

```

X2[t_] = solmake[A, λ, m, e2, t]
{e^{-2t} (-1 - 2t), e^{-2t} \left(-t^2 + 2 \left(t + \frac{t^2}{2}\right)\right), 4 e^{-2t} t, e^{-2t} (2(1-t) - 2t)}

X3[t_] = solmake[A, λ, m, e3, t]
{e^{-2t} \left(-1 - 2t - t^2 + 2 \left(t + \frac{t^2}{2}\right)\right), 0, e^{-2t} (2(1-t) + 2t), 0}

```

```
X4[t_] = solmake[A, λ, m, e4, t]
```

```
{e-2 t (-1 - 2 t), e-2 t (1 + 2 t), 4 e-2 t t, -4 e-2 t t}
```

Our general solution is

```
Xgen[t_] = a1 * X1[t] + a2 * X2[t] + a3 * X3[t] + a4 * X4[t]
```

```
{a1 + a2 e-2 t (-1 - 2 t) +
  a4 e-2 t (-1 - 2 t) + a3 e-2 t (-1 - 2 t - t2 + 2 (t +  $\frac{t^2}{2}$ )),
  a1 + a4 e-2 t (1 + 2 t) + a2 e-2 t (-t2 + 2 (t +  $\frac{t^2}{2}$ )),
  4 a2 e-2 t t + 4 a4 e-2 t t + a3 e-2 t (2 (1 - t) + 2 t),
  a2 e-2 t (2 (1 - t) - 2 t) - 4 a4 e-2 t t}
```

As an example, we solve an initial value problem.

```
Xinit = {0, 0, 2, 2};
```

```
ans = Solve[Xgen[0] == Xinit, {a1, a2, a3, a4}]
```

```
{{a1 → 1, a2 → 1, a3 → 1, a4 → -1}}
```

```
Xsole[t_] = Xgen[t] /. Thread[Flatten[ans]]
```

```
{1 + e-2 t (-1 - 2 t - t2 + 2 (t +  $\frac{t^2}{2}$ )),
  1 - e-2 t (1 + 2 t) + e-2 t (-t2 + 2 (t +  $\frac{t^2}{2}$ )),
  e-2 t (2 (1 - t) + 2 t), e-2 t (2 (1 - t) - 2 t) + 4 e-2 t t}
```

```
Xsole[t_] = Simplify[%]
```

```
{1 - e-2 t, 1 - e-2 t, 2 e-2 t, 2 e-2 t}
```

Let's check both the differential equations and the initial conditions.

```
D[Xsole[t], t] - A.Xsole[t]
```

```
{0, 0, 0, 0}
```

```
Xsole[0] - Xinit
```

```
{0, 0, 0, 0}
```

■ Example 4

As a final example, we will consider a system with repeated complex eigenvalues.

```
A = {{0, 1, 0, 0}, {0, 0, 1, 0}, {0, 0, 0, 1}, {-25, 20, -14, 4}}
{{0, 1, 0, 0}, {0, 0, 1, 0}, {0, 0, 0, 1}, {-25, 20, -14, 4}}
```

```
Eigenvalues[A]
```

```
{1 - 2 i, 1 - 2 i, 1 + 2 i, 1 + 2 i}
```

We have two eigenvalues of multiplicity 2, both complex.

```
m = 2;
```

```
ndim = 4;
```

```
 $\lambda = 1 + 2 I;$ 
```

```
M = MatrixPower[A -  $\lambda$  * Id, m]
```

```
{{-3 + 4 i, -2 - 4 i, 1, 0}, {0, -3 + 4 i, -2 - 4 i, 1},
{-25, 20, -17 + 4 i, 2 - 4 i}, {-50 + 100 i, 15 - 80 i, -8 + 56 i, -9 - 12 i}}
```

```
ans = NullSpace[M]
```

```
{{22 - 4 i, 15 + 20 i, 0, 125}, {-9 - 12 i, 10 - 20 i, 25, 0}}
```

```
e1 = ans[[1]]
```

```
{22 - 4 i, 15 + 20 i, 0, 125}
```

```
e2 = ans[[2]]
```

```
{-9 - 12 i, 10 - 20 i, 25, 0}
```

```
Z1[t_] = solmake[A,  $\lambda$ , m, e1, t]
```

```
{e(1+2 i) t ((22 - 4 i) (1 - (1 + 2 i) t) + (15 + 20 i) t),
(15 + 20 i) e(1+2 i) t (1 - (1 + 2 i) t), 125 e(1+2 i) t t,
e(1+2 i) t ((-250 + 500 i) t + 125 (1 + (3 - 2 i) t))}
```

```
Z2[t_] = solmake[A, λ, m, e2, t]
```

```
{e(1+2 i) t ((-9 - 12 i) (1 - (1 + 2 i) t) + (10 - 20 i) t),
 e(1+2 i) t ((10 - 20 i) (1 - (1 + 2 i) t) + 25 t),
 25 e(1+2 i) t (1 - (1 + 2 i) t), (75 - 100 i) e(1+2 i) t t}
```

We get from this four real-valued solutions by taking real and imaginary parts. To do this cleanly, we need to be able to tell *Mathematica* that t is real. In *Mathematica* 3, this requires a package called Algebra ReIm, which allows us to declare quantities as real. In *Mathematica* 4, it is much simpler. We may make such a declaration in a simplify statement.

```
Clear[X1, X2, X3, X4];
```

```
X1[t_] = Simplify[Re[ComplexExpand[Z1[t]]], t ∈ Reals]
```

```
{et ((22 - 15 t) Cos[2 t] + 4 (1 + 5 t) Sin[2 t]),
 5 et ((3 + 5 t) Cos[2 t] + 2 (-2 + 5 t) Sin[2 t]),
 125 et t Cos[2 t], 125 et ((1 + t) Cos[2 t] - 2 t Sin[2 t])}
```

```
X2[t_] = Simplify[Im[ComplexExpand[Z1[t]]], t ∈ Reals]
```

```
{-et (4 (1 + 5 t) Cos[2 t] + (-22 + 15 t) Sin[2 t]),
 -5 et (2 (-2 + 5 t) Cos[2 t] - (3 + 5 t) Sin[2 t]),
 250 et t Cos[t] Sin[t], 125 et (2 t Cos[2 t] + (1 + t) Sin[2 t])}
```

```
X3[t_] = Simplify[Re[ComplexExpand[Z2[t]]], t ∈ Reals]
```

```
{-et ((9 + 5 t) Cos[2 t] + 2 (-6 + 5 t) Sin[2 t]),
 -5 et ((-2 + 5 t) Cos[2 t] - 4 Sin[2 t]),
 -25 et ((-1 + t) Cos[2 t] - 2 t Sin[2 t]), 25 et t (3 Cos[2 t] + 4 Sin[2 t])}
```

```
X4[t_] = Simplify[Im[ComplexExpand[Z2[t]]], t ∈ Reals]
```

```
{et (2 (-6 + 5 t) Cos[2 t] - (9 + 5 t) Sin[2 t]),
 -5 et (4 Cos[2 t] + (-2 + 5 t) Sin[2 t]),
 -25 et (2 t Cos[2 t] + (-1 + t) Sin[2 t]),
 -25 et t (4 Cos[2 t] - 3 Sin[2 t])}
```